
COMP 410
Summer 2024

Midterm Practice Exam #1 (Solutions)

This is representative of the kinds of topics and kind of questions you may be asked on the

midterm. This practice exam, along with assignments 1-3 and the 6in-class handouts, are 
intended to be comprehensive of everything on the exam. That is, I will not ask anything that’s 
not somehow covered by those sources.  (I will announce the exact cutoff for the handouts on 
Wednesday.)


You are permitted to bring two 8.5 x 11 sheets of paper into the exam with you, as long as they

have handwritten notes on them. Both sides of both sheets can be used. To be clear, these

must be entirely handwritten.


Abstract Syntax Trees

In Boolean expressions, ¬ has the highest precedence, followed by ∧ and ∨.  With this 
in mind, write out the ASTs corresponding to each of the following Boolean expressions:

1.) ¬a ∧ b ∨ c



2.) (a ∨ b) ∧ c



3.) ¬(a ∧ b) ∧ (b ∨ c) 

 



Arithmetic expressions can be used to form Boolean expressions with the help of 
arithmetic comparisons (e.g., <, <=, >, >=, ==).  These comparisons have the lowest 
possible precedence.  With this in mind, write out the ASTs corresponding to each of the 
following expressions:

4.) 1 * 2 + 3 == 4



5.) (2 + 2 < 4) ∧ ¬a 

 



6.) Consider the following Python class definitions, which are adapted from assignment 
1's boolean evaluator.  These classes are used to represent AST nodes.

class And: 
  def __init__(self, left, right): 
    self.left = left 
    self.right = right 

class Or: 
  def __init__(self, left, right): 
    self.left = left 
    self.right = right 

Assume that Boolean true is represented as an AST with Python's True, and Boolean 
false is represented as an AST with Python's False.  With all this in mind, represent the 
following Boolean expressions in Python using And, Or, True, and False as 
appropriate.

6.a) true ∧ false

And(True, False) 

6.b.) false ∨ true

Or(False, True) 

6.c.) false ∧ true ∨ true

Or(And(False, True), True) 

6.d.) false ∨ true ∧ true

Or(False, And(True, True)) 



Semantic Tableau

For each of the following Boolean formulas, write out the complete semantic tableau 
tree.  Circle the nodes in the tree representing solutions.  If a tree has no solutions, say 
so.  Be sure to write all steps.

7.) ¬a ∧ a



8.) (a ∨ ¬a) ∧ a 

 



9.) (¬x ∧ ¬y) ∨ (x ∧ y) 

 



Prolog - Modeling the World

10.a)
For this problem, you need to write a clause database encapsulating pricing information 
for a convenience store.  Write Prolog code accurately reflecting the following:

• Soda costs $2
• Chips cost $3
• Hot dogs cost twice as much as soda (do not hardcode $4)
• Soda chips, and hot dogs are food
• Pencils and pens are office supplies
• All office supplies cost $2
• Cold medicine costs $7

% all facts and rules with the same name should be placed 
% together in the file 
cost(soda, 2). 
cost(chips, 3). 
cost(hot_dog, Cost) :- 
    cost(soda, SodaCost), 
    Cost is SodaCost * 2. 
cost(OS, 2) :- 
    office_supplies(OS). 
cost(cold_medicine, 7). 

food(soda). 
food(chips). 
food(hot_dog). 

office_supplies(pencil). 
office_supplies(pen). 



Using the clause database you previously wrote, write queries to determine the 
following:

10.b.) Which items cost exactly $2?

?- cost(Item, 2). 

10.c.) Which items cost more than $3?

?- cost(Item, Cost), Cost > 3. 

10.d.) Which foods cost less than $3?

?- food(Food), cost(Food, Cost), Cost < 3. 

10.e.) Which foods are also office supplies?

?- food(Item), office_supplies(Item). 



Unification

Consider each of the following unification attempts.  If the unification succeeds, record 
any values any variables take.  If the unification fails, say so.

11.) foo(1, X) = foo(Y, 2)

X = 2, Y = 1 

12.) foo(1, X) = foo(X, 2)

false 

13.) foo(1, _) = foo(X, 2)

X = 1 

14.) foo(1, _) = foo(1, _)

true 

15.) foo(1, 2, bar) = foo(X, _, _, _)

false 

16.) foo(bar(baz), X) = foo(Y, Z), Y = bar(A)

X = Z, Y = bar(baz), A = baz 



Recursion

17.) Consider the following mathematical definition of a recursive function:

Write an equivalent definition in Prolog.

f(0, 2). 
f(1, 3). 
f(N, Result) :- 
    N > 1, 
    MinOne is N - 1, 
    MinTwo is N - 2, 
    f(MinOne, T1), 
    f(MinTwo, T2), 
    Result is (3 * T1) + (4 * T2). 

18.) Write a procedure named evensBetween, which will nondeterministically produce 
all the even numbers within an inclusive range.  As a hint, a number N is even if and 
only if the clause 0 is mod(N, 2) is true.  An example query is below:

?- evensBetween(1, 4, Even). 
Even = 2 ; 
Even = 4. 

evensBetween(Min, Max, Min) :- 
    Min	 =< Max, 
    0 is mod(Min, 2). 
evensBetween(Min, Max, Result) :- 
    Min	 < Max, 
    NewMin is Min + 1, 
    evensBetween(NewMin, Max, Result). 

fn =

⎧

⎪

⎨

⎪

⎩

2 if n = 0

3 if n = 1

(3× fn−1) + (4× fn−2) otherwise



Unification with Lists

Consider each of the following unification attempts involving lists.  If the unification 
succeeds, record any values any variables take.  If the unification fails, say so.

19.) [1, 2, _] = [A, B, C|D]

A = 1, B = 2, D = [] 

20.) A = [1, 2|B], B = [4]

A = [1, 2, 4], B = [4] 

21.) [[A|B], C] = [[1, 2]|D]

A = 1, B = [2], D = [C] 

22.) X = [A|[2]]

X = [A, 2] 

23.) [A, [B, [C|D]]] = [1, [2, [3, 4]]]

A = 1, B = 2, C = 3, D = [4] 



Consider the following inductive list definition, which makes use of Prolog atoms and 
structures:

Now consider the following unifications, using Prolog lists.  Rewrite these unifications 
using the above definition.

24.) X = [1, 2, 3]

X = cons(1, cons(2, cons(3, nil))) 

25.) X = [Y|Z]

X = cons(Y, Z) 

26.) X = [A|[2]]

X = cons(A, cons(2, nil)) 

27.) X = [1, [2, [3]]]

X = cons(1, cons(cons(2, cons(cons(3, nil), nil)), nil)) 



More Recursion

28.) Write a procedure named allEqual which will succeed if all list elements are 
equal to each other according to unification (=).  You may introduce any helpers you 
wish.  Example calls are below:

?- allEqual([]). 
true. 
?- allEqual([1, 1, 1]). 
true. 
?- allEqual([1, 2, 3]). 
false. 
?- allEqual([1, X, 1]). 
X = 1. 
?- allEqual([A, B]). 
A = B.  
?- allEqual([X, 1, 2]). 
false. 

allEqual([]). 
allEqual([_]). 
allEqual([H, H|Rest]) :- 
    allEqual([H|Rest]). 



29.) Write a procedure named zip, which takes two lists of the same length, an output 
list of the same length.  The output list is a list of pair structures, where each pair 
holds an element from each list, preserving order.  If the lists are not the same length, 
zip should fail, though you shouldn't need to explicitly check the length.  Example calls 
are below:

?- zip([], [], Output). 
Output = []. 
?- zip([hello], [goodbye], Output). 
Output = [pair(hello, goodbye)]. 
?- zip([1, 2, 3], [a, b, c], Output). 
Output = [pair(1, a), pair(2, b), pair(3, c)]. 
?- zip([A, B], [C, D], Output). 
Output = [pair(A, C), pair(B, D)]. 
?- zip([foo], [bar, baz], Output). 
false. 
?- zip([foo, bar], [baz], Output). 
false. 

zip([], [], []). 
zip([H1|T1], [H2|T2], [pair(H1, H2)|Rest]) :- 
    zip(T1, T2, Rest). 


